

LINKING GTAP TO NATIONAL MODELS: SOME HIGHLIGHTS AND A PRACTICAL APPROACH

Mark Horridge¹

Joaquim Bento de Souza Ferreira Filho²

1 Introduction

The GTAP model is designed for analysis related to international trade. As such, the model comprises a detailed specification of trade between countries, as well as a full set of accounts for the domestic economies. For many situations or types of policy analysis, however, more detailed modeling of specific national economies may be required. This is the case, for example, of questions demanding inter-regional analysis inside the national economies.

Among the many possible interesting applications of the GTAP model, then, is its use in conjunction to national models. The interest in this kind of application is growing in recent years, and encompasses a wide range of theoretical as well as practical questions. This paper provides some hints and considerations on linking the GTAP model to national, more detailed models, as well as a practical implementation linking GTAP to a inter-regional version of ORANIGFR, calibrated with Brazilian data. In what follows, we first set the main difficulties regarding that linkage. Then the general differences between a global model like GTAP and a single country model are analyzed, in order to highlight the more convenient ways of transmitting the shocks. And finally, an example is provided of linking GTAP to a national model, and the results of this new methodology are compared to previous approaches.

2 The nature of the problem

Consider the case of a large country involved in an economic trade block creation, and in which the economic activity is unequally spread across the national territory. In this case, a relevant problem could be not just whether the integration would affect the country positively or not, but also how would regions inside the

¹ Centre of Policy Studies – COPS, Monash University, Melbourne, Australia. Email: Mark.Horridge@BusEco.monash.edu.au

² Escola Superior de Agricultura “Luiz de Queiroz”, University of Sao Paulo, Piraciaba ,SP, Brazil. Email: jbsferre@esalq.usp.br . This paper was written while the author was a visiting scholar at COPS, Monash University. Financial aid of FAPESP is gratefully acknowledged.

country be affected. This type of question, however, can't be addressed in the standard GTAP framework.

One possible solution, then, would be to have a national model with a detailed inter-regional specification at national level, suitable for the type of question proposed. The national model could be used in conjunction with GTAP in such a way that the integration impacts over the national economy would be generated running GTAP, and the consequent shocks would be transmitted to the national economy model for the inter-regional analysis.

This kind of application is potentially very promising, but encompasses a series of difficult questions. The first problem that immediately arises is how to do the transmission of the shocks between the two models. This is both a theoretical and an empirical issue. From the theoretical side, we note that GTAP relates all economies in a particular regional aggregation through a set of export and import demand curves. The solution for a simulation is a set of import and export prices and quantities, and the choice of each of them to select to impose on the national model is not unique. One could, for example, choose to impose to the national model the import prices and the export quantities generated by GTAP, and let the national model give the correspondent import quantities and export prices. Or else, the import quantities and the export prices could be imposed to the national model, that would then calculate the correspondent import prices and export quantities. But it would not be possible, in general, to impose all the values above mentioned. And from the empirical side there remains the problem of how to do the job.

The second and more difficult question is whether the shocks generated by both models are consistent one to each other. In other words, is the equilibrium result generated in the national model consistent with the one generated for the national economy by GTAP? This aspect of the problem can be clarified further with an example. Consider that one chooses to impose to the national model the import price and the export quantity changes generated by the GTAP model. The national model, then, will generate new import quantities and export price changes. Are these import quantities and export price changes consistent with the initial import prices and export quantity changes (shocks) generated by GTAP?

To this point, it should be clarified what is meant by "consistent" here. In practice, it is common that national models do not have exactly the same database as the GTAP model, since it is easier for the national models to have more updated

databases than for the GTAP, a global model. Yet, even when the year databases are the same there is room for discrepancies between them, due to different criteria used in its construction. All these differences turn it difficult to refer to the level of the variables in comparing results. Our consistency check, then, refers to the percent changes in the variables, arising from both models.

3 Some theoretical considerations on the model structure

In order to highlight the channels through which the shocks between models should be transmitted, in this section we analyze the main differences in the structure of a single country model and a world model as GTAP. We will focus here on the most general features, and in those that are more relevant to the question at hand. Among the many differences one can expect between models, then, we will focus on the interface between a global model and a single country model.

As stated above, GTAP links all countries through a set of import and export demand curves. The iteration between them determines a new equilibrium for prices and quantities of goods in the world. In GTAP, then, there are no exogenous commodities prices, since the model is global. Any price or quantity change in any country inside the model must generate changes in the overall world equilibrium, and a new equilibrium must be computed.

In a single country model, however, a much different adjustment process takes place. Typically, a single country model takes as given the external prices of both its exports and imports. This is a very important difference between models, and a major point regarding the problem under discussion. Let's for the moment consider the export demand in the single country model with more detail. Generally speaking, the demand for exports in single country models usually follows the well-known Armington type specification:

$$X_4(c) = F_4Q(c) \cdot \left[\frac{P_4(c)}{F_4P(c) \cdot PHI} \right]^{EXP_ELAST(c)} \quad (1)$$

In the equation above, $X_4(c)$ and $P_4(c)$ are commodities export quantities and domestic prices, EXP_ELAST is the export elasticity (a negative number), PHI is the nominal exchange rate, and $F_4Q(c)$ and $F_4P(c)$ are shifters in the export demand schedule. $F_4P(c)$ can be interpreted as the world price of each good. If the domestic price P_4 increases in relation to the world prices, then export quantities should fall.

$F_4Q(c)$ is a quantity shifter in export demand that summarizes a wide range of possible variable changes: changes in tastes, in world income, and any other events that could exogenously move the export demands in the country. Note that F_4Q and F_4P would be typically set as exogenous in the closure of the single model.

One important point to be noted here, then, is that in this formulation the only way of increasing exports is to decrease its prices in relation to the world prices. Export quantities and prices must be inversely related. This is not, however, necessarily the case in GTAP, where an export quantity rise in any particular country can be associated both to an increase or a decrease in its export prices, depending on the kind of shock given to the model. This happens due to the more general adjustment required in a global model, where the world price is required to change.

This kind of adjustment is easily observed in trading blocks formation experiments, where the simultaneous change in import tariffs that may be much different across countries can generate that result, due to the elimination of trade distortions caused by the tariffs. For the same world price change, some countries may experience export quantities increases, while others will face export quantities decreases. For instance, in a GTAP simulation for the FTAA formation we can observe, for the GTAP lum (wood products) sector, an increase in pxw (the aggregate export price index) both for Brazil (1.20%) and the Rest of Mercosur (2.30%), but a decrease (-1.39%) in qxw (the aggregate exports) for Brazil, and an increase (0.69%) for the Rest of Mercosur. How, then, to cope with this fact in the single country model?

The answer is that in order to deal with all possible changes in export prices and quantities, it is necessary to adjust the positions of the export demand curves in the single country model, in order to accommodate the cases where export prices and quantities move in the same direction. In what follows, we discuss this point in details, and show the results of linking the GTAP model to the ORANIGFR model calibrated for Brazil.

4 Linking GTAP to a single country model

4.1 The ORANIGFR model

The ORANIGFR is a model developed by the Centre of Policy Studies, Monash University, Melbourne, Australia. It is a well-known static AGE model, with many

different versions and applications, both in Australia and in other countries in the world. It is built in the linearization tradition, as the GTAP model, what makes it suitable for the task at hand.

4.2 The Variables to Be Linked

Let's get back to the trade block formation example, where a group of countries decide to eliminate all the barriers to trade among them. This will be called the trade liberalization experiment. At the same time, we want to link the results to a different single country model of one of the integrating countries. We'll need here in many instances to particularize the single country to each we refer. For that purpose and due also to the example to be used in this paper, the single country under consideration will be Brazil, for which a shortcut "BRA" will be adopted throughout.

The first thing to note is what should be transmitted. Focusing initially in the export side of the problem, we see that the relevant variables in GTAP are $p_{fob}(i,r,s)$, the aggregate FOB (percent change) export price index of commodity i from region r to region s , and $q_{xs}(i,r,s)$, the respective percent change in export quantity. Rewriting equation (1) above in percent change form, we get:

$$x_4(c) - f_4q(c) = EXP_ELAST(c) * [p_4(c) - \phi - f_4p(c)] \quad (2)$$

where, as traditionally stated, the lower case variables stand for percent variations. Here, $EXP_ELAST(c)$ is a negative number.

An additional relation important for the discussion is:

$$p_4(c) = p_e(c) + t_4(c) \quad (3)$$

which just states the zero pure profits in exports in the single country model, where $p_e(c)$ is the domestic basic price of exports, and $t_4(c)$ is the export tax.

The variable p_{fob} is the international FOB price of goods exported by region r , the equivalent of p_4 in equation (2)³. At the same time, the variable q_{xs} , the export quantity, must also be linked to the variable x_4 . The linking process then must provide a way of linking $p_4(c)$ to $p_{fob}(i,"BRA",s)$, and $x_4(c)$ to $q_{xs}(i,"BRA",s)$. Note that the variables p_4 and x_4 don't have the subscript "BRA", since they already refer to the Brazilian single country model.

In performing the linkage discussed above, two problems immediately arise. Firstly, the variables in the single country model are vectors of commodity prices and

³ Note that f_4p would be equivalent to composite world price of the commodity, an aggregation of the commodity export price of every region at world prices.

quantities, while in the global model they are matrices. At this point, it could be argued why not to use $p_{fob}(i, "BRA")$ and $q_{xw}(i, "BRA")$, the aggregated export price and quantities of each region, instead of p_{fob} and q_{xs} . The reason is that this formulation (that is possible) would not allow the analysis of trade liberalization issues, since the type of shock required usually comprises uneven changes in tariffs between regions. That would be impossible to do if just the aggregated p_{xw} and q_{xw} were to be used. Using p_{fob} and q_{xs} , then, will require some type of aggregation procedure.

Secondly, it should be noted that both $p_4(c)$ and $x_4(c)$ are endogenous to the national model. If their values are to be required to be equal to any other value determined outside the national model, as is the case if we link those values to the GTAP variables values, then some mechanism must be put into action in order to allow the linkage. The "natural" guess here would be, naturally, to endogenize f_{4q} , the exogenous shift variable in the export demand function of the single country model. As discussed before f_{4q} can be regarded as a summary of many possible different types of changes in the export demand functions, not explicitly taken into account by the single country model. If f_{4q} varies endogenously, it will accommodate the changes in x_4 necessary to guarantee the consistency between the two models.

The change in f_{4q} then will be equivalent to the required shift in the export demand schedules in order to have x_4 matching $q_{xs}(i, "BRA", s)$. It can be interpreted as is a kind of "shadow" variable, showing the equivalent movement in the single country's exports demand schedules associated with each respective pair p_{fob} and q_{xs} . From a single country model perspective, these changes cannot, generally speaking, be accessed without the help of a global model. With the procedure described above it is possible to impose the values of both p_{fob} and q_{xs} generated from GTAP to the single country model.

And finally it should be noted that if p_4 is chosen to be linked to p_{fob} directly, it will not be possible to perform region specific shocks to the export taxes. Although this may be quite unusual, it must be useful in modeling Voluntary Exports Restraints, for example. Due to this, the variable to be linked to p_{fob} will be $pe(c)$, showed above in equation (3), in the single country model. In this case, $t_4(c)$ must also be linked to $[tx(i, "BRA") + txs(i, "BRA", s)]$, the export tax variables in the GTAP model.

The problem, as discussed so far, just comprised the single country perspective. Let's now discuss the problem from the global model perspective, still in the export

side of the question. From the GTAP perspective, the problem is more or less the same. The variables $qxs(i, "BRA", s)$ and $pfob(i, "BRA", s)$ are endogenous to the GTAP model. If they are to be set to another exogenously calculated value (as is the case with the national model), some mechanism must be provided in order to allow that link. Actually, the problem in hand could be summarized as finding a way of "turning off" the Brazilian part of the GTAP model, and "plugging in" the Brazilian national model in its place. Since the regions in GTAP are connected to each other through the export and import demand functions, it seems that these functions must be "turned off" in Brazilian part of GTAP. This can be done with the aid of special SLACK variables, placed appropriately in the code. These variables, once turned endogenous, would allow the link between the models. This is the general solution to the question, which will become clearer when we provide the details in the next sections.

In the import side, the problem is symmetric to what was discussed so far. In the national model, the world import prices are exogenous. This variable, in ORANIGFR is called $pf0cif(c)$, a vector of commodity CIF prices in external currency. The correspondent import quantities are called $x0imp(c)$. The $pf0cif(c)$ in the national model, then, should be linked to the $pcif(i, r, "BRA")$ in GTAP, the CIF world price of commodity i supplied from every region r to Brazil. Here, the same rationale used in using $pfob$ instead of pxw applies regarding using $pcif$ instead of $piw(i, s)$. And, again, $pcif$ is a matrix, and some kind of aggregation must be provided.

Note that $pf0cif(c)$ is already exogenous to the national single country model, what means that linking it to $pcif$ does not require, from the national model perspective, any additional changes but endogenizing it. The variable $pcif(i, r, "BRA")$ is determined by GTAP, in all regions that trade with Brazil. Once $pf0cif(c)$ is determined, the single country model determines $x0imp(c)$, which must now be linked to $qim(i, "BRA")$ in GTAP. This is done through placing another SLACK variable in the Brazilian part of the GTAP code, what will have the effect of "turning off" the import demands in that part of the global model. In its place the link will plug the ORANIGFR import demand equations, that will transmit the value of $x0imp(c) = qim(i, "BRA")$ to the other regions in GTAP.

And, finally, it must be stressed that since the trade liberalization shock will be performed in GTAP, the global model, we must also provide a link between $t0imp(c)$, the import tariffs in the single country model, and $tms(i, r, "BRA")$, the Brazilian import tariffs in GTAP. When the region specific trade liberalization shocks to tms

are implemented in GTAP, the Brazilian single country model (ORANIGFR) must be able to translate these shocks into $t0imp(c)$, in order to capture correctly the tax collection effect. The variable $t0imp(c)$, then, that is originally exogenous, must be endogeneized in the Brazilian national model and set equal to an aggregation of $tms(i,r,"BRA")$ in GTAP.

The Table 1 summarizes the variable relations discussed so far. A more detailed table will be provided after the discussion of model implementation, to be done in the next section.

Table 1. Variables to be linked in the GTAP and ORANIGFR models.

	LINKS	
	ORANIGFR	GTAP
Import prices	$pf0cif(i)$	$pcif(i,r,"BRA")$
Import quantities	$x0imp(i)$	$qim(i,"BRA")$
Import tariffs	$t0imp(i)$	$tms(i,r,"BRA")$
Export prices	$pe(i)$	$Pfob(i,"BRA",r)$
Export quantities	$x4(i)$	$qxs(i,"BRA",r)$
Export taxes	$t4(i)$	$[tx(i,"BRA") + txs(i,"BRA",s)]$

After the linking of these variables, the Brazilian part of GTAP must be turned off, and the national model (ORANIGFR) must be turned on. The figure bellow shows an illustration of the process.




Figure 1. Scheme of the linking process.

4.3 Model Implementation

In the linking process, both models were initially aggregated to the same number of commodities, through a mapping that is showed in the Appendix 1. The number of industries need not be the same in both models, since all the linking happens at the commodity level. The model linking strategy pursued in this paper consists in integrating both models in one TABLO file. For this purpose, the ORANIGFR model code was pasted in the bottom of the GTAP model code. At this stage, we must be sure that there is no coincidence between names of variables, coefficients, sets, etc, in the two models, for what the TABLO check device in TABMATE is essential. Once this kind of semantic problem is cleared, the next stage consists in the linking strategy itself. In what follows, the specific code needed to perform the linking is displayed and commented. It is split in several excerpts, for easy of exposition.

The **Excerpt lk_1** shows some auxiliary coefficients required to implement the linking. The main purpose of these coefficients is to allow a mapping between the GTAP and the ORANIGFR sets, since they are not required to have the same names. Then, for example, the set commodities in ORANIGFR is the set COM, while in GTAP is the set TRAD_COMM. The names of the elements in each set are also different, and the mapping device provided by the above coefficients avoids the need to change them. Note that these coefficients are diagonal matrices, with the Brazilian imports and exports values in the main diagonal, and that the row labels have the ORANIGFR COM set names, while the column labels are the GTAP TRAD_COMM set elements names. These coefficients are written to a summary file for diagnostic purposes.

```
!*****LINKING GTAP TO ORANIGFR MODULE*****!
```

```
!Excerpt lk_1: Auxiliary coefficients for mapping purposes!
```

```
Coefficient
```

```
(all,c,COM)(all,g,TRAD_COMM)(all,r,REG) DIAG(c,g,r)#Diagonal matrix with 1's#;  
(all,c,COM)(all,g,TRAD_COMM)(all,r,REG)  
IMPORTS(c,g,r) #Coefficient mapping TRAD_COMM TO COM in the import side#;  
(all,c,COM)(all,g,TRAD_COMM)(all,r,REG)  
EXPORTS(c,g,r) #Coefficient mapping TRAD_COMM TO COM in the export side#;
```

```
Formula
```

```
(all,c,COM)(all,g,TRAD_COMM)(all,r,REG) DIAG(c,g,r)=0.0;  
(all,c,COM)(all,g,TRAD_COMM)(all,r,REG:$POS(c)=$POS(g)) DIAG(c,g,r) = 1;  
(all,c,COM)(all,g,TRAD_COMM)(all,r,REG)  
IMPORTS(c,g,r) = DIAG(c,g,r)*VIWS(g,r,"BRA");  
(all,c,COM)(all,g,TRAD_COMM)(all,r,REG)  
EXPORTS(c,g,r) = DIAG(c,g,r)*VXWD(g,"BRA",r);
```

Write
 IMPORTS to file summary header "IMPS";
 EXPORTS to file summary header "EXPS";
 VIWS to file summary header "VIWS";
 VXWD to file summary header "VXWS";
 DIAG to file summary header "DIAG";

The **Excerpt lk_2** shows the new variables needed for the linking equations. As will be seen below the role of these variables is to create the link between the two models. If they are endogenous, the link is turned off, since they will absorb the linking effects. If they are exogenous, then the link is turned on, and the ORANIGFR variables and the GTAP variables are effectively linked.

!Excerpt lk_2: Linking variables !

Variable

(all,c,COM) slack1(c);
 (all,c,COM) slack2(c);
 (all,g,TRAD_COMM)(all,r,REG) slack3(g,r);
 (all,g,TRAD_COMM)(all,s,REG) slack4(g,s);

The **Excerpt lk_3** shows the linking between the export taxes and the import tariffs in both models. The role of the coefficients EXPORTS and IMPORTS can be seen here. The equation E_LINKt4 sets that t4(c) must be equal to the weighted aggregation of the Brazilian export taxes in GTAP, the weights being the Brazilian exports. Note that both the coefficients EXPORTS and IMPORTS are updated, since they are derived from VIWS(i,r,s) and VXWD(i,r,s).

!Excerpt lk_3: Export taxes and import tariffs linking equations!

Equation LINK_t4

(all,c,COM) [sum{g,TRAD_COMM,sum{s,REG,EXPORTS(c,g,s)}}]*t4(c) =
 - sum{g,TRAD_COMM,sum{s,REG,EXPORTS(c,g,s)}*
 [tx(g,"BRA") + txs(g,"BRA",s)}}];

Equation LINK_t0imp

(all,c,COM) [sum{g,TRAD_COMM,sum{r,REG,IMPORTS(c,g,r)}}]*t0imp(c) =
 [sum{g,TRAD_COMM,sum{r,REG,IMPORTS(c,g,r)*tms(g,r,"BRA")}}];

The **Excerpt lk_4** shows the linking equations in the models. The equation names were chosen to explicitly highlight its role in the system. Below each equation there is a comment that shows the target to be match by one. In the equation LINK_pf0cif_pcif, then, the variable pf0cif must be turned endogenous in order to

perform the link. As seen before, slack1(c) is the switch that turns this equation on (if exogenous) or off (if endogenous). In order to turn the link on, then, one must swap pf0cif with slack1 in the closure.

!Excerpt lk_4: Linking equations!

!Variables determined by GTAP!

Equation LINK_pf0cif_pcif #Equation linking pf0cifreg in ORANIGFR to pcif IN GTAP#

```
(all,c,COM) [sum{g,TRAD_COMM,sum{r,REG,IMPORTS(c,g,r)}}]*pf0cif(c)=
    slack1(c)+ [sum{g,TRAD_COMM,sum{r,REG,IMPORTS(c,g,r)*pcif(g,r,"BRA")}}];
! pf0cif !
```

The next equation links x4 to qxs(i,"BRA",s). The slack variable in this equation is slack2(c), which must be exogenous when the link is on. The comment bellow the equation shows that the variable f4q(c) in ORANIGFR must be endogenized when the link is on. Swapping f4q(c) with slack2(c), then, activates the link between the models for this equation.

Equation LINK_x4_qxs #Equation linking x4 in ORANIGFR to qxs IN GTAP#

```
(all,c,COM) [sum{g,TRAD_COMM,sum{r,REG,EXPORTS(c,g,r)}}]*x4(c)=
    slack2(c)+ [sum{g,TRAD_COMM,sum{r,REG,EXPORTS(c,g,r)*qxs(g,"BRA",r)}}];
! f4q !
```

The equation Link_pfob_pe links the export prices in both models. The slack variable here is slack3(g,s), and the comment line bellow the equation reminds us that its target is the variable lk_slack3(i,"BRA",r). This a new variable, that must be inserted in the GTAP code to "turn off" the Brazilian GTAP export demand equation, as discussed before.

!Variables determined by ORANIGFR!

Equation LINK_pfob_pe #Equation linking pfob in GTAP to pe in ORANIGFR#

```
(all,g,TRAD_COMM)(all,s,REG)
[sum{c,COM,EXPORTS(c,g,s)}+TINY]*[pfob(g,"BRA",s) + tx(g,"BRA")+txs(g,"BRA",s)]
= slack3(g,s) + sum{c,COM,[EXPORTS(c,g,s)+TINY]*[pe(c)-phi]};
! lk_slack3(i,BRA,r) !
```

The correspondent change in the GTAP code is as follows. First, create the new variable:

```
Variable (all,i,TRAD_COMM)(all,r,REG)(all,s,REG) lk_slack3(i,r,s) #Variable linked to slack3 in
the linking equations#;
```

Then, the new variable must be inserted in the equation bellow:

```
Equation EXPRICES # eq'n links agent's and world prices (HT 27) #
(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    pfob(i,r,s) = pm(i,r) - tx(i,r) - txs(i,r,s) + lk_slack3(i,r,s);
```

The variable $lk_slack3(i,r,s)$ will be exogenous when the link is off, and endogenous when the link is on. The variable $slack3$, on the other hand, will be endogenous when the link is off, and exogenous when the link is on. To turn the link on, then, the variable $lk_slack3(TRAN_COMM, "BRA", REG)$ must be swapped with $slack3(g,s)$.

And, finally, the equation $LINK_qim_x0imp$ provides the link between the import quantities in the models. The comment bellow the equation shows another new variable, $lk_slack4(i)$, which must be created in the GTAP code. The variable $lk_slack4(i)$ will be exogenous when the link is off, and endogenous when the link is on. The variable $slack4$, on the other hand, will be endogenous when the link is off, and exogenous when the link is on. To turn the link on, then, the variable $lk_slack4(TRAN_COMM)$ must be swapped with $slack4(i)$. The piece of code bellow shows the linking equations in $ORANIGFR$.

```
Equation LINK_qim_x0imp #Equation linking qim(i,s) in GTAP to x0imp in ORANIGFR#
(all,g,TRAD_COMM)
    ID01[sum{c,COM,sum{r,REG,IMPORTS(c,g,r)}}]*qim(g,"BRA") = slack4(g) +
        sum{c,COM,sum{r,REG,IMPORTS(c,g,r)*x0imp(c)}};
!(all,i,TRAD_COMM) lk_slack4(i) !
```

The associated new GTAP variable and equations are as follows:

```
Set Country # region modeled by non-GTAP model # (BRA);
Subset Country is subset of REG;
Coefficient (all,r,REG) IsCountry(r) # binary dummy: 1=special region #;
Formula (all,r,REG) IsCountry(r)=0;
Formula (all,r,Country) IsCountry(r)=1;
Variable (all,i,TRAD_COMM) lk_slack4t(i) # Equation to be linked to slack4 in ORANIGFR#;
Equation GHHLDAGRIMP # government consumption demand for aggregate imports (HT 43) #
(all,i,TRAD_COMM)(all,s,REG)
    qgm(i,s) = qg(i,s) + ESUBD(i) * [pg(i,s) - pgm(i,s)] - IsCountry(s)*lk_slack4(i);
Equation PHHLDAGRIMP # private consumption demand for aggregate imports. (HT 49) #
(all,i,TRAD_COMM)(all,s,REG)
    qpm(i,s) = qp(i,s) + ESUBD(i) * [pp(i,s) - ppm(i,s)] - IsCountry(s)*lk_slack4(i);
```

Equation INDIMP # industry j demands for composite import i (HT 31) #

(all,i,TRAD_COMM)(all,j,PROD_COMM)(all,s,REG)

$$qfm(i,j,s) = qf(i,j,s) - ESUBD(i) * [pfm(i,j,s) - pf(i,j,s)] - IsCountry(s)*lk_slack4(i);$$

The variable IsCountry is a dummy variable with values 1 for the country under focus (Brazil), and 0 for the remaining countries. Its purpose, then, is to allow the link to be done to just one country, while keeping the code general. Again, the situation is symmetric as that discussed before regarding lk_slack3.

The Table 2 shows a summary of the changes required in the code to link the models.

Table 2. Variables and swaps in the linked model.

Equation	Variables		Slacks		SWAPS
	ORANIGFR	GTAP	ORANIGFR	GTAP	
LINK_pf0cif_pcif	pf0cif(c)	pcif(i,r,"BRA")	slack1		pf0cif = slack1
LINK_x4_qxs	x4(c)	qxs(i,"BRA",r)	slack2		f4q = slack2
LINK_pfob_pe	pe(c)	pfob(i,"BRA",r)	slack3	lk_slack3	lk_slack3 = slack3
LINK_qim_x0imp	x0imp(c)	qim(i,"BRA")	slack4	lk_slack4	lk_slack4 = slack4
LINK_t0imp	t0imp(c)	tms(g,r,"BRA")	endogenize t0imp		
LINK_t4	t4(c)	[tx(g,"BRA") + txs(g,"BRA",s)]	endogenize t4		

As noted before, the Brazilian part of GTAP is now meaningless. This means that the “numéraire” definition in GTAP must also be changed, in order to disregard pfactor(“BRA”) in the Brazilian part. This is done in the code bellow:

Coefficient

VENDWWLD # value of primary factors, at mkt prices, worldwide #;

Formula

VENDWWLD = sum(r,REG,[1-IS-COUNTRY(r)]*VENDWREG(r));

Variable (orig_level = 1.0) pfactwld # world price index of primary factors #;

Equation PRIMFACTPRWLD # computes % change in global price index of primary factors #

VENDWWLD * pfactwld = sum(r,REG,[1-IS-COUNTRY(r)]*VENDWREG(r) * pfactor(r));

Note that now the “numéraire” definition in GTAP does not use the Brazilian market price index of primary factors. The GTAP “numéraire”, then, is not affected by the values of the slack variables introduced. And note also that ORANIGFR has its own “numéraire”, the nominal exchange rate (denoted by PHI in the single country model.)

Some final comments must be done about the proposed method of linking the models. First, note that the Brazilian region in the GTAP model is completely

deactivated by the use of the slacks. The Brazilian part of tms, tx and txs, then, does not perform any role in the model anymore. But the link between them and the correspondent taxes in ORANIGFR allows us to shock them as they are, since the shocks will be automatically transmitted from one model to the other.

And finally, note that the model here presented is tailored for trade liberalization shocks. Other kinds of shocks would require other links. A technological change shock to be applied to every region, for example, would require the linking of the appropriate technological change variables in the models, in the same way that was done for the taxes, "mutatis mutandis".

5 Comparing results between the suggested approach and previous approaches

In order to highlight the differences in results between the approach suggested here and previous approaches to the problem, some results will be showed. We performed a comparative simulation where the results of the suggested methodology were compared with another approach previously used (Adams et alii, 1988). In Adams et alii (1988) approach the simulation (a trade liberalization experiment) is run in GTAP in first place, and its results are transmitted to the national model through shocks to it in a second round. The models, then, are run separately. The variables used to transmit the shocks between the models were the import tariffs, the import prices and the export prices. As seen before, the latter variable is normally endogenous to national models. The shock to it was implemented shocking a price shifter in the export demand function.

The simulation performed was the creation of the Free Trade Area of Americas, with a full elimination of import tariffs between the member countries. The aggregation used comprised 7 regions, and 34 commodities, in a one to one correspondence in both models, and 34 industries⁴. The Brazilian single country used is an inter-regional version of ORANIGFR, with 27 regions distinguished inside Brazil, calibrated with 1996 data. The inter-regional system is modeled "top-down".

⁴ The number of commodities and industries need not be the same in the new national model, since all the linking process is done at the commodity level.

The next figures bring some selected results about the simulations. Figure 2 shows the differences in the Brazilian export prices in both models. As it can be seen, there are significant differences among them, including some export prices inverse variations.




Figure 2. Brazilian export prices comparison: linked and non-linked models.




Figure 3. Brazilian export quantities comparison: linked and non-linked models.

The reasons for the difference in the results are straightforward. In the non-linked model the export price shifter is shocked by 1.22%, the same variation produced in the Brazilian export prices in the previous GTAP run. This is equivalent to a right shift in the Brazilian export demand schedule, enough to generate a 4.07% increase in the Brazilian exports in the non-linked model.

In the linked model the trade liberalization shock and the variables linkage generates a Brazilian export demand shifter variation of 2.17%. But now this variation is not enough to generate export demands increases, but an export demand fall of 5.84%. This is caused by a simultaneous left shift in the export supply schedule, more intense in the linked model than in the non-linked one. The effective price of the composite primary factor input increased by 1.32% in the linked model, and by 0.3% in the non-linked model.

The next figures bring results from the import side.




Figure 4. Brazilian import prices comparison: linked and non-linked models.




Figure 5. Brazilian import quantities comparison: linked and non-linked models.

As can be seen in the figures above, models results differences seem to be smaller in the import side than in the export side. The results over domestic variables can be seen in the next figures.




Figure 6. Brazilian output of commodities comparison: linked and non-linked models.

As it can be seen from the figure above, the two models give very different results for the domestic production of commodities. And finally, some inter-regional results generated by the Brazilian single country model are presented.




Figure 7. Regional employment, regional wage bills and regional gross products. Brazil.

Figure 7 above shows the % variation in regional employment (regemploy_i), the regional wage bills (regw1lab_io) and gross regional products for the 27 Brazilian states. As can be seen from the figure, the FTAA would change the distribution of employment in Brazil, increasing the employment in some of the peripheral states of the poorest regions in Brazil, the Northeast region (states of Piaui, Ceara, Rio Grande do Norte, Paraiba, Alagoas and Sergipe).

6 Final Remarks

The method here proposed allows the linking of a single country model to the GTAP model. The linking methodology suggested is general enough to be extended for other single country models. The example here provided used the same number of commodities and industries in both models, but this is not required. All the linkage process happens at commodity level, what allows the single country model to have a different number of industries (industries with multi-production technology, for instance).

The number of commodities also need not be the same, but different number of commodities would require a mapping between commodities in both models. We have however used the one to one correspondence in this paper and due to this reason we prefer to refer to this last aspect as work in progress.

And finally it should be stressed that the linking process here suggested substitutes part of the GTAP model by a different model, granting consistency between the commodity flows in international trade generated by both models. This new model, of course, will behave differently from the other regions in the world, with specific assumptions regarding factor mobility, production, and so one, and these differences should be taken into account for results interpretation. Particular attention should be paid to the kind of shock being implemented in the model, since this will affect the choice of the variables to be linked.

7 References

ADAMS, P.D; HORRIDGE, M; PARMENTER, B; ZHANG, X.G. Long Run Effects on China of APEC Trade Liberalization. General Paper no. G-130. Centre of Policy Studies. The Impact Project. Monash University. Melbourne, October, 1998.

ACKNOWLEDGEMENTS

The authors would like to thank the participants of a seminar presented at the Centre of Policy Studies – COPS, Monash University, for their comments and suggestions to a former version of this paper. The remaining errors and omissions are of our own responsibility.

APPENDIX 1: Correspondence between commodities classification in GTAP and in ORANIGFR

	Products and sectors	
	GTAP	ORANIGFR
1	agp	Agriculture
2	omn	MineralExtr
3	cog	PetrGasExtr
4	nmm	MinNonMet
5	i s	IronProduc
6	nfm	MetalNonFerr
7	fmp	OtherMetal
8	otn	OthMachPart
9	ele	EletricMat
10	ome	EletronEquip
11	mvh	Automobiles

12	lum	WoodFurnit
13	ppp	PaperGraph
14	crp	ChemicElem
15	p_c	PetrolRefin
16	tex	Textiles
17	wap	Apparel
18	lea	ShoesInd
19	pcr	VegetProcess
20	cmt	Slaughter
21	mil	Dairy
22	sgr	SugarInd
23	vol	VegetOils
24	omf	VariousInd
25	egw	PubUtilServ
26	cns	CivilConst
27	trd	Trade
28	trp	Transport
29	cmn	Comunic
30	fsi	FinancInst
31	ros	FamServic
32	obs	EnterpServ
33	dwe	BuildRentals
34	osg	PublAdm